# next ordinal in a class

## Johan G. F. Belinfante
## 2010 February 5

```
In[1]:= SetDirectory["l:"]; << goedel.10feb04a;<< tools.m

:Package Title: goedel.10feb04a                    2010 February 4 at 7:55 p.m.

It is now:  2010 Feb 5 at 12:31

Loading Simplification Rules

TOOLS.M                                 Revised 2010 January 29

weightlimit = 40
```

## summary

The function **ordlist[x]** is either a finite list or a countably infinite sequence that lists ordinals (if any) in a class **x** in increasing order, starting with the least ordinal in the class **x**. If there are no ordinals in the class **x**, then **ordlist[x] = 0**. A specific example of interest is the sequence **PRIMESEQ** of prime natural numbers.

```
In[2]:= ordlist[PRIMES]

Out[2]= PRIMESEQ
```

The definition of **ordlist[x]** involves an **iterate** construction:

```
In[3]:= iterate[composite[HULL[intersection[OMEGA, x]], SUCC], set[A[intersection[OMEGA, x]]]]

Out[3]= ordlist[x]
```

The starting point of this list of ordinals is the least ordinal in **x**, if any. (If there are no ordinals in **x**, then this expression reduces to **V**.)

```
In[4]:= APPLY[ordlist[x], 0]

Out[4]= A[intersection[OMEGA, x]]
```

The recursion relation for **ordlist[x]** involves the function **HULL[Ω ⋂ x] ∘ SUCC** that takes each ordinal to the next ordinal in the sequence.

```
In[5]:= composite[ordlist[x], SUCC]

Out[5]= composite[HULL[intersection[OMEGA, x]], SUCC, ordlist[x]]
```

In this notebook is shown that this next-ordinal function is the cover relation of the well-founded relation **id[Ω ⋂ x] ∘ E**. If has already been shown that the cover relation of this well-founded relation is a function.

```
In[6]:= FUNCTION[cover[composite[id[intersection[OMEGA, x]], E]]]

Out[6]= True
```

Moreover, it is already known that the next ordinal function and the cover relation have the same domain:

```
In[7]:= composite[HULL[intersection[OMEGA, x]], SUCC] // domain

Out[7]= U[intersection[OMEGA, x]]

In[8]:= cover[composite[id[intersection[OMEGA, x]], E]] // domain

Out[8]= U[intersection[OMEGA, x]]
```

Thus, to show they are equal, it suffices to show that one of them is contained in the other. Moreover, it is already known at this point that the next ordinal function is a subclass of the well-founded relation.

```
In[9]:= subclass[composite[HULL[intersection[OMEGA, x]], SUCC],
         composite[id[intersection[OMEGA, x]], E]]

Out[9]= True
```

The cover relation of a well-founded relation **wf[x]** is given by

```
In[10]:= dif[wf[x], composite[wf[x], wf[x]]]

Out[10]= cover[wf[x]]
```

Therefore to show that the next ordinal function is the cover relation of the well-founded relation **id[Ω ⋂ x] ∘ E** it suffices to show that the next-ordinal function is disjoint from the composite of this well-founded relation with itself.

---

## derivation

Lemma.

```
In[11]:= Map[empty[composite[Id, complement[#]]] &,
           SubstTest[class, pair[y, z], or[not[member[y, z]], not[member[z, x]],
              not[subclass[x, t]], not[member[z, hull[x, succ[y]]]]]],
             t → OMEGA] /. x → intersection[OMEGA, t]] // InvertFix

Out[11]= equal[0, intersection[OMEGA, t, fix[composite[inverse[E],
              BIGCUP, IMAGE[HULL[intersection[OMEGA, t]]], IMAGE[SUCC]]]]] == True

In[12]:= intersection[OMEGA, t_, fix[composite[inverse[E],
             BIGCUP, IMAGE[HULL[intersection[OMEGA, t_]]], IMAGE[SUCC]]]] := 0
```

If **fix[u ∘ v] = 0**, then also **fix[v ∘ u] = 0**. This observation yields the following corollary.

Corollary.

```
In[13]:= SubstTest[implies, empty[fix[composite[u, v]]], empty[fix[composite[v, u]]],
           {u → composite[id[intersection[OMEGA, t]], inverse[E]], v -> composite[
              inverse[E], IMAGE[HULL[intersection[OMEGA, t]]], IMAGE[SUCC]]}] // Reverse

Out[13]= equal[0, fix[composite[inverse[E], IMAGE[HULL[intersection[OMEGA, t]]],
            IMAGE[SUCC], BIGCUP, IMAGE[id[intersection[OMEGA, t]]]]]] == True

In[14]:= fix[composite[inverse[E], IMAGE[HULL[intersection[OMEGA, t_]]],
            IMAGE[SUCC], BIGCUP, IMAGE[id[intersection[OMEGA, t_]]]]] := 0
```

The fixed point set of a composite is the range of an intersection. So, if this fixed point class is empty, then so is that intersection. This observation yields a reformulation without **fix**.

Corollary.

```
In[15]:= SubstTest[composite, id[range[t]], t,
          t -> intersection[composite[HULL[intersection[OMEGA, x]], SUCC],
            composite[inverse[IMAGE[id[intersection[OMEGA, x]]]], inverse[BIGCUP], E]]]

Out[15]= intersection[composite[HULL[intersection[OMEGA, x]], SUCC],
           composite[inverse[IMAGE[id[intersection[OMEGA, x]]]], inverse[BIGCUP], E]] == 0

In[16]:= intersection[composite[HULL[intersection[OMEGA, x_]], SUCC],
           composite[inverse[IMAGE[id[intersection[OMEGA, x_]]]], inverse[BIGCUP], E]] := 0
```

Lemma. (A technical lemma needed to cope with a rewrite rule for complements of composites involving functions.)

```
In[17]:= SubstTest[subclass, u, complement[v],
           {u -> composite[HULL[intersection[OMEGA, x]], SUCC], v -> composite[
              inverse[IMAGE[id[intersection[OMEGA, x]]]], inverse[BIGCUP], E]}] // Reverse

Out[17]= subclass[composite[HULL[intersection[OMEGA, x]], SUCC], composite[
            inverse[IMAGE[id[intersection[OMEGA, x]]]], inverse[BIGCUP], complement[E]]] == True

In[18]:= (% /. x → x_) /. Equal → SetDelayed
```

Lemma. The next-ordinal function is a subclass of the cover relation of **id[Ω ⋂ x] ∘ E**.

```
In[19]:= Map[subclass[composite[HULL[intersection[OMEGA, x]], SUCC], #] &, SubstTest[dif,
           wf[t], composite[wf[t], wf[t]], t → composite[id[intersection[OMEGA, x]], E]]]

Out[19]= subclass[composite[HULL[intersection[OMEGA, x]], SUCC],
           cover[composite[id[intersection[OMEGA, x]], E]]] == True

In[20]:= (% /. x → x_) /. Equal → SetDelayed
```

Theorem. The next-ordinal function is equal to the cover relation of **id[Ω ⋂ x] ∘ E**.

```
In[21]:= SubstTest[implies, and[subclass[u, v], FUNCTION[v]],
           equal[u, composite[v, id[domain[u]]]]],
           {u -> composite[HULL[intersection[OMEGA, x]], SUCC],
            v -> cover[composite[id[intersection[OMEGA, x]], E]]}] // Reverse

Out[21]= equal[composite[HULL[intersection[OMEGA, x]], SUCC],
           cover[composite[id[intersection[OMEGA, x]], E]]] == True

In[22]:= cover[composite[id[intersection[OMEGA, x_]], E]] :=
          composite[HULL[intersection[OMEGA, x]], SUCC]
```

Corollary. (Replace $\Omega \bigcap \mathbf{x}$ wrapper with $\mathbf{x} \subset \Omega$ literal.)

```
In[23]:= SubstTest[implies, equal[x, intersection[OMEGA, t]],
           equal[cover[composite[id[x], E]], composite[HULL[x], SUCC]], t → x] // Reverse

Out[23]= or[equal[composite[HULL[x], SUCC], cover[composite[id[x], E]]],
           not[subclass[x, OMEGA]]] == True

In[24]:= or[equal[composite[HULL[x_], SUCC], cover[composite[id[x_], E]]],
           not[subclass[x_, OMEGA]]] := True
```

## the prime sequence

The function **PRIMESEQ** which lists all the primes in increasing order is an example of an **ordlist** function. This infinite list starts with **APPLY[PRIMESEQ, 0] = 2**, and continues forever. Note that the recursion relation for the prime sequence involves the next-prime function **HULL[PRIMES]** ∘ **SUCC**.

```
In[25]:= composite[PRIMESEQ, SUCC]

Out[25]= composite[HULL[PRIMES], SUCC, PRIMESEQ]
```

Corollary. The next-prime function is the cover relation of **id[PRIMES]** ∘ **E**.

```
In[26]:= SubstTest[cover, composite[id[intersection[OMEGA, x]], E], x → PRIMES] // Reverse

Out[26]= cover[composite[id[PRIMES], E]] == composite[HULL[PRIMES], SUCC]

In[27]:= cover[composite[id[PRIMES], E]] := composite[HULL[PRIMES], SUCC]
```